RISC-V REPORT

Abdelrahman Atef Youssef
Submitted to: Eng. Sara
Design Track

Table of Contents

Architecture:



Codes & Block diagrams:

1) mux_2:

RISC-V > risc-v > mux_2.v

```
1  module mux_2 (
2      A, B, sel, MUXOut
3  );
4      input [31:0] A, B;
5      input sel;
6      output wire [31:0] MUXOut;
7      ////////////////////////////////////////
8      assign MUXOut = (sel == 1'b0) ? A : B;
9  endmodule
```



2) add_by_4:

RISC-V > risc-v > add_by_4.v

```
1  module add_by_4 (
2      PC, PCPlus4
3  );
4      input [31:0] PC;
5      output wire [31:0] PCPlus4;
6      ////////////////////////////////////////
7      assign PCPlus4 = PC + 32'h0004;
8  endmodule
```



3) imm_ext:

RISC-V > risc-v > imm_ext.v

```
1  //This models the unit necessary to extend the immediate field//
2  ////////////////////////////////////////////////////////////////
3  module imm_ext (
4      Instr, ImmSrc, ImmExt
5  );
6      input [31:7] Instr;
7      input [1:0] ImmSrc;
8      output reg [31:0] ImmExt;
9      ////////////////////////////////////////////////////////
10     always @(Instr or ImmSrc) begin
11         ImmExt[31:12] = {20{Instr[31]}};
12         case (ImmSrc)
13             2'b00: ImmExt[11:0] = {Instr[31:20]};
14             2'b01: ImmExt[11:0] = {Instr[31:25], Instr[11:7]};
15             2'b10: ImmExt[11:0] = {Instr[7], Instr[30:25], Instr[11:8], 1'b0};
16             2'b11: ImmExt[19:0] = {Instr[19:12], Instr[20], Instr[30:21], 1'b0};
17             default: ImmExt = 32'h0000;
18         endcase
19     end
20 endmodule
```



4) add_to_ImmExt:

RISC-V > risc-v > add_to_ImmExt.v

```
1  module add_to_ImmExt (
2      A, B, Sum
3  );
4      input [31:0] A, B;
5      output wire [31:0] Sum;
6      ////////////////////////////////////
7      assign Sum = A + B;
8  endmodule
```



5) PC:

RISC-V > risc-v > PC.v

```
1   //This models the Program Counter unit
2   ////////////////////////////////////////////////////
3   module PC (
4       clk, reset, PC, PCNext
5   );
6       input clk, reset;
7       input [31:0] PCNext;
8       output reg [31:0] PC;
9       ////////////////////////////////////
10      always @(posedge clk or posedge reset) begin
11          if (reset == 1'b1)
12              PC <= 32'h0000;
13          else
14              PC <= PCNext;
15      end
16  endmodule
```



6) reg_file:

RISC-V > risc-v > reg_file.v

```
1   //Triple port register file
2   //2 ports to read from, 1 to write to
3
4   module reg_file (
5       A1, A2, A3, clk, WE3, WD3, RD1, RD2
6   );
7       input [4:0] A1, A2, A3;
8       input clk, WE3;
9       input [31:0] WD3;
10      output wire [31:0] RD1, RD2;
11      ////////////////////////////////////////
12      reg [31:0] rf [31:0];
13      ////////////////////////////////////////
14      always @(posedge clk) begin
15          rf[0] = 32'h0000;
16          if (WE3 == 1'b1)
17              if (A3) rf[A3] <= WD3;
18      end
19      assign RD1 = rf[A1];
20      assign RD2 = rf[A2];
21  endmodule
```



7) instr_mem:

RISC-V > risc-v > instr_mem.v

```
1   //Single port ROM used to store instructions only//
2   ////////////////////////////////////////////////////
3   module instr_mem (
4       PC, Instr
5   );
6       input [31:0] PC;
7       output wire [31:0] Instr;
8       ////////////////////////////////////////////////
9       reg [31:0] instr_mem [255:0];
10      ////////////////////////////////////////////////
11      initial begin
12          $readmemh ("program.s", instr_mem);
13      end
14      assign Instr = instr_mem[PC[9:2]];
15  endmodule
```

8) data_mem:

RISC-V > risc-v > data_mem.v

```
1   //Single port RAM used to store data only (accessed word by word)
2   ////////////////////////////////////////////////////////////////////
3   module data_mem (
4       clk, WE, WD, A, RD
5   );
6       input clk, WE;
7       input [31:0] WD;
8       input [31:0] A;
9       output wire [31:0] RD;
10      ////////////////////////////////////////////////
11      reg [31:0] data_mem [255:0];
12      ////////////////////////////////////////////////
13      always @(posedge clk) begin
14          if (WE == 1'b1)
15              data_mem[A[9:2]] <= WD;
16      end
17      assign RD = data_mem[A[9:2]];
18  endmodule
```

9) alu:

RISC-V > risc-v > alu.v

```
1   //ALU module implements 4 main operations as follows:
2   //ALUCtrl operation
3   //  00        add
4   //  01        sub
5   //  10        and
6   //  11        or
7   ////////////////////////////////////////
8   module alu #(parameter N = 32)
9               (A, B, ALUCtrl, ALUResult, Zero);
10      input [N-1 : 0] A, B;
11      input [1:0] ALUCtrl;
12      output reg [N-1 : 0] ALUResult;
13      output wire Zero;
14      ////////////////////////////////////////
15      assign Zero = (ALUResult == 0) ? 1'b1 : 1'b0;
16      always @(A or B or ALUCtrl) begin
17          case (ALUCtrl)
18              2'b00:  ALUResult = A + B;
19              2'b01:  ALUResult = A - B;
20              2'b10:  ALUResult = A & B;
21              2'b11:  ALUResult = A | B;
22              default: ALUResult = 32'h0000;
23          endcase
24      end
25  endmodule
```



10) alu_decoder:

RISC-V > risc-v > alu_decoder.v

```
1   //Input: ALUOp, funct3(Instr[14:12]), opcode[5](Instr[5]), funct7[5](Instr[30])
2   //Output: ALUCtrl
3   module alu_decoder (
4       ALUOp, funct3, op_5, funct7_5, ALUCtrl
5   );
6       input [1:0] ALUOp;
7       input [2:0] funct3;
8       input op_5, funct7_5;
9       output reg [1:0] ALUCtrl;
10      //////////////////////////////////////////////////
11      always @(ALUOp, funct3, op_5, funct7_5) begin
12          case (ALUOp)
13              2'b00:  ALUCtrl = 2'b00;    //LW & JALR
14              2'b01:  ALUCtrl = 2'b01;    //SW
15              2'b10:  case ({funct3, op_5, funct7_5})
16                          5'b000_0_0, 5'b000_0_1, 5'b000_1_0: ALUCtrl = 2'b00;    //ADD & ADDI
17                          5'b000_1_1: ALUCtrl = 2'b01;                            //SUB
18                          5'b111_1_0: ALUCtrl = 2'b10;                            //AND & ANDI
19                          5'b110_1_0: ALUCtrl = 2'b11;                            //OR  & ORI
20                      endcase
21              default: ALUCtrl = 2'bxx;
22          endcase
23      end
24  endmodule
```



11) main_decoder:

RISC-V > risc-v > main_decoder.v

```v
//Input: opcode (Instr[6:0])
//Output: RegWrite, ImmSrc, ALUSrc, Branch, MemWrite, ResultSrc, ALUOp, Jump
module main_decoder (
    opcode, RegWrite, ImmSrc, ALUSrc, Branch, MemWrite, ResultSrc, ALUOp ,Jump
);
    input [6:0] opcode;
    output wire RegWrite, ALUSrc, Branch, MemWrite, ResultSrc, Jump;
    output wire [1:0] ImmSrc, ALUOp;
    ////////////////////////////////////////////////////////////
    reg [9:0] controls;
    ////////////////////////////////////////////////////////////
    assign {RegWrite, ImmSrc, ALUSrc, Branch, MemWrite, ResultSrc, ALUOp, Jump} = controls;
    always @(opcode) begin
        case (opcode)
            7'b0000011: controls = 10'b1_00_1_0_0_1_00_0;  //LW
            7'b0100011: controls = 10'b0_01_1_0_1_0_00_0;  //SW
            7'b1100011: controls = 10'b0_10_0_1_0_0_01_0;  //BEQ & BNE
            7'b0110011: controls = 10'b1_00_0_0_0_0_10_0;  //R-Type
            7'b0010011: controls = 10'b1_00_1_0_0_0_10_0;  //ADDI & ORI & ANDI
            7'b1101111: controls = 10'b1_11_0_0_0_0_00_1;  //JAL
            7'b1100111: controls = 10'b1_00_1_0_0_0_00_1;  //JALR
            default: controls = 10'bxxxxxxxxxx;
        endcase
    end
endmodule
```



12) ctrl_unit:

RISC-V > risc-v > ctrl_unit.v

```
1  module ctrl_unit (
2      opcode, funct3, funct7_5, Zero, ImmSrc, ALUCtrl,
3      RegWrite,
4  ALUSrc, MemWrite, ResultSrc, PCSrc, Jump
5  );
6      input [6:0] opcode;
7      input [2:0] funct3;
8      input funct7_5, Zero;
9      output wire [1:0] ImmSrc, ALUCtrl;
10     output wire RegWrite, ALUSrc, MemWrite, ResultSrc, PCSrc, Jump;
11     ////////////////////////////////////////////////////////
12     wire [1:0] ALUOp;
13     wire Branch;
14     ////////////////////////////////////////////////////////
15     assign PCSrc = (Zero & Branch & ~funct3[0]) | (~Zero & Branch & funct3[0]) | Jump;
16     //Instantiate building blocks
17     main_decoder main_decoder (opcode, RegWrite, ImmSrc, ALUSrc, Branch, MemWrite, ResultSrc, ALUOp, Jump);
18     alu_decoder alu_decoder (ALUOp, funct3, opcode[5], funct7_5, ALUCtrl);
19 endmodule
```



13) datapath:

RISC-V > risc-v > datapath.v

```
module datapath (
    clk, reset, RegWrite, ALUSrc, PCSrc, ResultSrc, Jump,
    Instr, ALUCtrl, ImmSrc, RD, PC, ALUResult, RD2, Zero
);
    input clk, reset, RegWrite, ALUSrc, PCSrc, ResultSrc, Jump;
    input [31:0] Instr;
    input [1:0] ALUCtrl, ImmSrc;
    input [31:0] RD;
    output wire [31:0] PC;
    output wire [31:0] ALUResult;
    output wire [31:0] RD2;
    output wire Zero;
    ////////////////////////////////////////////
    wire [31:0] RD1, WD3, PCTarget, PCPlus4, PCNext, ImmExt, TmpResult, TmpPC;
    wire [31:0] SrcB;
    //Instantiate building blocks
    add_by_4 add_by_4 (.PC(PC), .PCPlus4(PCPlus4));
    ////////////////////////////////////////////////////////////////////////////
    add_to_ImmExt add_to_ImmExt (PC, ImmExt, TmpPC);
    ////////////////////////////////////////////////////////////////////////////
    PC pc (.clk(clk), .reset(reset), .PC(PC), .PCNext(PCNext));
    ////////////////////////////////////////////////////////////////////////////
    reg_file reg_file (.A1(Instr[19:15]), .A2(Instr[24:20]), .A3(Instr[11:7]),
                       .WE3(RegWrite), .RD1(RD1), .RD2(RD2), .WD3(WD3), .clk(clk));
    ////////////////////////////////////////////////////////////////////////////
    imm_ext imm_ext (Instr[31:7], ImmSrc, ImmExt);
    ////////////////////////////////////////////////////////////////////////////
    mux_2 i1 (RD2, ImmExt, ALUSrc, SrcB);
    ////////////////////////////////////////////////////////////////////////////
    alu alu (.A(RD1), .B(SrcB), .ALUCtrl(ALUCtrl), .ALUResult(ALUResult), .Zero(Zero));
    ////////////////////////////////////////////////////////////////////////////
    mux_2 i2 (ALUResult, RD, ResultSrc, TmpResult);
    ////////////////////////////////////////////////////////////////////////////
    mux_2 i3 (TmpResult, PCPlus4, Jump, WD3);
    ////////////////////////////////////////////////////////////////////////////
    mux_2 i4 (TmpPC, ALUResult, (Jump & ALUSrc), PCTarget);
    ////////////////////////////////////////////////////////////////////////////
    mux_2 i5 (PCPlus4, PCTarget, PCSrc, PCNext);
    ////////////////////////////////////////////////////////////////////////////

endmodule
```



14)top:

RISC-V > risc-v > top.v

```
module top (
    clk, reset
);
    input clk, reset;
    ////////////////////////////////////////////////////////////////////////////////////////////////////////////////
    wire [31:0] PC, Instr, ALUResult, RD, WD;
    wire RegWrite, Zero, MemWrite, ResultSrc, ALUSrc, PCSrc, Jump;
    wire [1:0] ImmSrc, ALUCtrl;
    reg [8*4 : 1] op_literal;
    ////////////////////////////////////////////////////////////////////////////////////////////////////////////////
    datapath datapath (clk, reset, RegWrite, ALUSrc, PCSrc, ResultSrc, Jump, Instr, ALUCtrl, ImmSrc, RD, PC, ALUResult, WD, Zero);
    ////////////////////////////////////////////////////////////////////////////////////////////////////////////////
    ctrl_unit ctrl_unit (Instr[6:0], Instr[14:12], Instr[30], Zero, ImmSrc, ALUCtrl, RegWrite, ALUSrc, MemWrite, ResultSrc, PCSrc, Jump);
    ////////////////////////////////////////////////////////////////////////////////////////////////////////////////
    instr_mem instr_mem (PC, Instr);
    ////////////////////////////////////////////////////////////////////////////////////////////////////////////////
    data_mem data_mem (clk, MemWrite, WD, ALUResult, RD);
    ////////////////////////////////////////////////////////////////////////////////////////////////////////////////
endmodule
```



15)tb:

RISC-V > risc-v > tb.v

```
1   module tb;
2       reg clk, reset;
3       ////////////////////////////////////////
4       top top (clk, reset);
5       ////////////////////////////////////////
6       localparam PERIOD = 10;
7       ////////////////////////////////////////
8       initial begin
9           clk = 1'b0;
10          forever begin
11              #(PERIOD/2) clk = ~clk;
12          end
13      end
14      initial begin
15          reset = 1'b1;
16          #6  reset = 1'b0;
17      end
18  endmodule
```

Testbench:

Program (1):

RISC-V > risc-v > sample.s

```
1   main:
2       addi x2, x0, 5 # x2 = 5 0 00500113
3       addi x3, x0, 12 # x3 = 12 4 00C00193
4       addi x7, x3, -9 # x7 = (12 - 9) = 3 8 FF718393
5       or x4, x7, x2 # x4 = (3 OR 5) = 7 C 0023E233
6       and x5, x3, x4 # x5 = (12 AND 7) = 4 10 0041F2B3
7       add x5, x5, x4 # x5 = 4 + 7 = 11 14 004282B3
8       beq x5, x7, end # shouldn't be taken 18 02728863
9       addi x4, zero, 0 # x4 = 0 1C 0041A233
10      beq x4, x0, around # should be taken 20 00020463
11      addi x5, x0, 0 # shouldn't execute 24 00000293
12  around:
13      addi x4, zero, 1 # x4 = 1 28 0023A233
14      add x7, x4, x5 # x7 = (1 + 11) = 12 2C 005203B3
15      sub x7, x7, x2 # x7 = (12 - 5) = 7 30 402383B3
16      sw x7, 84(x3) # [96] = 7 34 0471AA23
17      lw x2, 96(x0) # x2 = [96] = 7 38 06002103
18      add x9, x2, x5 # x9 = (7 + 11) = 18 3C 005104B3
19      jal x3, end # jump to end, x3 = 0x44 40 008001EF
20      addi x2, x0, 1 # shouldn't execute 44 00100113
21  end:
22      add x2, x2, x9 # x2 = (7 + 18) = 25 48
23      sw x2, 0x20(x3) # [100] = 25 0221A023
24  done:
25      beq x2, x2, done # infinite loop 50 00210063
```

Wavefroms:

Note: In the figures below, the waveform shows some internal signals of interest accompanied with “reg_file” and “data_mem” memories to facilitate observing changes to them. I tried to show up the intermediate changes through several snapshots of the simulation. Moreover, changes in both memories are highlighted so that it becomes hopefully easy to trace.




Msgs
reg_file
A1
A2
A3
clk
WE3
WD3
RD1
RD2
PC
PC
alu
A
B
ALUCtrl
ALUResult
Zero
Memory Data - /tb/top/data_mem/data_...
Memory Data - /tb/top/datapath...
File Edit View Bookmarks Window Help
Goto:
Now
Cursor 1


Msgs
reg_file
A1
A2
A3
clk
WE3
WD3
RD1
RD2
PC
PC
alu
A
B
ALUCtrl
ALUResult
Zero
Memory Data - /tb/top/data_mem/data_...
Memory Data - /tb/top/datapath...
File Edit View Bookmarks Window Help
Goto:
Now
Cursor 1
Cursor 2
104 ns to 156 ns
Now: 155 ns Delta: 4



Program (2):

This is a simple program that I wrote to calculate the factorial of a given number. Following is the source assembly code followed by simulation snapshots.

RISC-V > risc-v > factorial.s

```
1   addi s1, zero, 5     #x9 = 5 (Get 5!)
2   addi t0, zero, 0     #x5 = 0 (Constant for comparison)
3   addi t1, zero, 1     #x6 = 1 (Constant for comparison)
4   addi t2, s1, 0       #x7 = x9 (Final value of factorial to be stored in it)
5
6   factorial:
7       beq     s1, t0, done        #if (s1 == 0) -> done
8       beq     s1, t1, done        #if (s1 == 1) -> done
9       addi    s2, s1, -1          #x18 = x9 - 1 (s2 = s1 - 1)
10      beq     s2, t1, done        #if (s2 == 1) -> done
11      addi    t3, t2, 0           #x28 = x7 (Stores partial sums)
12      mul:
13          add  t2, t2, t3
14          addi s2, s2, -1
15          bne  s2, t1, mul
16          addi s1, s1, -1
17          j    factorial
18  done:
```

Example (1): Factorial of 5!



Example (2): Factorial of 6!